

## 11. 各種分割実験モデルに対するMIXED プロシジャーの活用

高橋 行雄

日本ロシュ(株)・臨床開発本部・統計解析室

Effective use for MIXED Procedure with Split-Plots Experimental Medels

Yukio Takahashi

Medical Biometrics CRD Div. Nippon Roche

要旨 リーリス 6.07 よりサポートされるようになった MIXED プロシジャーは、複数の誤差構造を持ち、かつ、欠測値を含むような分割実験、経時データなどの解析に多面的に活用できる。本質的に誤差が複数ある実験計画、変量因子を含む場合、単純な測定の繰り返し、実験の反復、多段分割、多方分割などについて MIXED プロシジャーの活用法を、特に分散の期待値の構造の観点より複数の誤差の合成について、GLM プロシジャーおよび VARCOMP プロシジャーと対比しながら解説した。

キーワード： SAS / MIXED, 分割実験, 変量因子, 混合モデル

### 1. はじめに

SAS の GLM プロシジャーは、一般化した解析方法により、重回帰分析、欠測値を含む各種の分散分析、共分散分析、数量化 I 類など、これまでの統計パッケージでは個別的に取り扱われてきた問題を統一化した。一般的に、統一的な手法というものは、使い勝手が悪いのであるが、GLM プロシジャーは、それを凌駕するパワーを持っていて、応用範囲の極めて広く、探索的なデータ解析の道具として欠かせない。しかしながら、分割実験のように誤差が複数あるような実験データの解析には、技工的な使い方が求められるにもかかわらず、複数の誤差を考慮しなければならない場合の水準平均および水準間の差の標準誤差について最高次の誤差しか用いないといった欠点がある。このために、結果が常に過大評価(有意差が出やすい)されやすく問題である。

さて、リーリス 6.07 よりサポートされるようになった MIXED プロシジャーは、複数の誤差構造を持ち、かつ、欠測値を含むような分割実験、経時データなどの解析に多面的に活用できる。ただし、GLM プロシジャー以上の汎用性があるがゆえに、これを適切に使いこなすことは容易ではない。そこで、従来の解析方法による様々な分割実験の型を示しつつ、MIXED プロシジャーの結果と対比し適切な使用方法を示したい。

## 2. 変量モデル

人を対象とした実験の場でしばしば問題となるのが"人"の取扱である。ランダムに選び出された5人のボランティアに3種類の薬物を、1週間の間隔でランダムに投与し、血圧の減少を計測したとしよう。結果を表示1に示す。

表示1 血圧の下降度 (単位 mmHg)

人	薬物 D <sub>1</sub>	薬物 D <sub>2</sub>	薬物 D <sub>3</sub>
R <sub>1</sub>	15	20	12
R <sub>2</sub>	9	14	7
R <sub>3</sub>	23	19	14
R <sub>4</sub>	10	17	8
R <sub>5</sub>	17	22	9

この実験モデルは、人をブロック(変量)とした乱塊法とみなせる。人を母数因子とみなそうと変量因子とみなそうと、薬物の効果は、繰り返しがないので人\*薬物の交互作用を誤差とみなして検定するしかない。ところが、薬物効果の平均値の標準誤差は、"人"を母数因子か変量因子かどちらにするのかによって異なる。母数因子とみなすと、この実験のボランティアについてのみの平均値の存在範囲の指標となり、変量因子とみなした場合は、全く別のボランティア5人についての平均値の存在範囲の指標となる。

MIXED プロシジャーは、人を母数因子とみなすか変量因子とみなすか、解析時にどちらでも簡単に対応できる。MIXED プロシジャーの使用方法は、GLM プロシジャーに VARCOMP プロシジャーの機能を組み込み、さらに機能拡張をしたものと言えよう。

MIXED プロシジャーを適切に使うためには、GLM プロシジャーおよび VARCOMP プロシジャーの結果と対比しながら解析を進めるとよい。表示2に GLM プロシジャーの RANDOM ステートメントにより変量因子を指定した結果を示す。変量因子Rと母数因子Dとの交互作用 R\*D も変量因子となるが、MODEL ステートメントに含めなければ自動的に誤差項とみなしてくれるので RANDOM ステートメントには含めないでおく。出力結果は、若干編集してあるが、分散分析表の分散の期待値の構造 Expected Mean Square から変量因子Rと母数因子Dともに誤差項で F 検定して良いことが示されている。問題は、LSMEANS ステートメントで推定されている標準誤差 Std Err である。これは、因子Rを母数因子とみなした結果で、MS の欄の誤差分散 V<sub>e</sub> = 6.5500 のみを用いて構成されている。因子Dの水準は5個のデータの平均であるので

$$SE_D = \sqrt{(V_e / n_D)} = \sqrt{(6.5500 / 5)} = 1.1445$$

として求められている。因子Rを変量因子とみなした標準誤差は、Var(R)の分散の期待値も誤差とみなして Var(Error)と合成する必要があるが、GLM プロシジャーでは対応して

表示2 GLM プロシジャーによる分散成分

```

data DD.D1 ;
  input R $ @ ;
  do D='D1', 'D2', 'D3' ;
    input y @ ;
    output ;
  end ;
  cards ;
    R1 15 20 12
    R2 9 14 7
    R3 23 19 14
    R4 10 17 8
    R5 17 22 9
  ;
proc GLM data=DD.D1 ;
  class R D ;
  model Y = R D / e1 ss1 ;
  random R ;
  lsmeans D / stderr ;
  estimate 'D1-D2' D 1 -1 0 ;
  estimate 'D1-D3' D 1 0 -1 ;
  estimate 'D2-D3' D 0 1 -1 ;
run ;

```

分散分析表 GLM Procedure

Source	DF	Type I SS	MS	F	Expected Mean Square
R	4	147.6000	36.9000	5.63	Var(Error) + 3 Var(R)
D	2	177.6000	88.8000	13.56	Var(Error) + Q(D)
Error	8	52.4000	6.5500		Var(Error)
Total	14	377.6000			

水準平均とSE Least Squares Means

D	Y	Std Err
D1	14.8000	1.1445
D2	18.4000	1.1445
D3	10.0000	1.1445

水準平均の差とSE

Paramater	Estimate	Std Err
D1-D2	-3.6000	1.6186
D1-D3	4.8000	1.6186
D2-D3	8.4000	1.6186

いないので、分散の期待値の構造を参照しながら手計算で求めなければならない。この例のようにデータが完備していれば Var(R) の係数は、整数となり簡単に求められるが、データが不揃いになった場合には、実数となり計算はかなり面倒となる。ただし、ESTIMTE ステートメントで求めている因子Dの水準間の差の SE は、いずれの場合でも Var(R) の影響を受けないので、Var(Error) のみを誤差とした出力は適切である。したがって、変量因子を含むような場合には、GLM プロシジャーには、このような問題が内在していることに注意しなければならない。

分散成分の推定値は、VARCOMP プロシジャーで手軽に求められる。表示3に示すように Type I の平方和(逐次方式)に基づく分散の期待値の構造を示す。MODEL ステートメ

ントで母数因子(固定効果)を先に列挙し、次に変量因子を列挙する。その境を FIXED オプションで指示する。この結果として  $\widehat{\text{Var}}(\text{R}) = 10.1167$  が得られる。これは、分散分析表の MS の欄より

$$\begin{aligned}\widehat{\text{Var}}(\text{R}) &= ((\widehat{\text{Var}}(\text{Error}) + 3 \widehat{\text{Var}}(\text{R})) - \widehat{\text{Var}}(\text{Error})) / 3 = (\widehat{V}_R - V_e) / 3 = \\ &= (36.9000 - 6.5500) / 3 = 10.1167\end{aligned}$$

として求めた結果と一致する。

表示3 VARCOMP プロシージャによる分散成分

proc varcomp data=DD.D1 method=type1 ; /* REPB01B.TXT */				
class D R ;				
model Y = D R / fixed=1 ;				
分散の期待値の構造				
Source	DF	Type I SS	MS	Expected Mean Square
D	2	177.600	88.800	$\text{Var}(\text{Error}) + Q(D)$
R	4	147.600	36.900	$\text{Var}(\text{Error}) + 3 \text{Var}(R)$
Error	8	52.400	6.550	$\text{Var}(\text{Error})$
Total	14	377.600		
Variance Component Estimate				
Var(R)		10.1167		
Var(Error)		6.5500		

表示4に MIXED プロシージャによる結果を示す。分散成分の推定は、制限付き最尤法 (REML) がデフォルトとなっている。分散の推定値は、Covariance Parameter Estimates の Estimate の欄に示されており、 $\widehat{\text{Var}}(\text{R})=10.1167$ ,  $\widehat{\text{Var}}(\text{Error})=\widehat{\text{Var}}(\text{Residual})=6.55$  と VARCOMP プロシージャの結果と一致している。母数因子DのF検定は、Tests of Fixed Effectsに示さ

表示4 MIXED プロシージャによる解析

proc MIXED data=DD.D1 ; /* REPB01C.TXT */				
class D R ;				
model Y = D ;				
random R ;				
lsmeans D / cl ;				
The MIXED Procedure				
Covariance Parameter Estimates (REML)				
Cov Parm	Ratio	Estimate	Std Error	Z Pr >  Z
R	1.5445	10.1167	8.7657	1.15 0.2484
Residual	1.0000	6.5500	3.2750	2.00 0.0455
Tests of Fixed Effects				
Source	NDF	ddf	Type III F	Pr > F
D	2	8	13.56	0.0027 **
Least Squares Means				
Level	LSMEAN	Std Error	ddf	Lower Upper
D D1	14.8000	1.8257	8	10.5898 19.0102
D D2	18.4000	1.8257	8	14.1898 22.6102
D D3	10.0000	1.8257	8	5.7898 14.2102

れている。検定結果は、Type III の F 値 13.56 が表示されていて、GLM プロシジャーの F 値と同じ結果になっている(完備型データなので Type I と同じになる)。因子 D の水準平均の標準誤差は、1.8257 となっていて、因子 R の分散を次の様に誤差とみなした結果となっていることが確認できる。

$$SE_D = \sqrt{(\widehat{\text{Var}(R)} + \widehat{\text{Var}(\text{Error})}) / n_D} = \sqrt{(10.1167 + 6.5500) / 5} = 1.8257$$

### 3. 分割実験での効果の推定

因子 R を実験の反復、因子 A を 1 次単位、因子 B を 2 次単位とする反復のある 1 段分割実験の例を表示 5 に示す。因子 R を変量因子とみなすことにより、前節と同様の考え方で解析できる。

表示 5 反復のある 1 段分割実験データ

	反復 R1				反復 R2			
	B1	B2	B3	B4	B1	B2	B3	B4
A1	12.3	13.9	15.0	15.0	11.7	12.7	13.0	13.4
A2	15.5	14.8	15.2	14.4	13.9	13.3	14.3	13.6
A3	13.9	13.3	14.3	12.8	14.0	13.2	13.5	12.8

GLM プロシジャーで交互作用 R\*A を 1 次誤差、交互作用 R\*B と R\*A\*B を 2 次誤差となるような解析事例を表示 6 に示す。反復 R と因子 A は、TEST ステートメントで 1 次誤差 R\*A を指定し直して検定しているが、誤差の自由度が小さいこともあり、共に有意な差ではない。因子 B および交互作用 A\*B は 2 次誤差に対して有意な差となる。

交互作用 A\*B は、1 次単位の因子 A と 2 次単位の因子 B の組み合わせなので、母平均に対する誤差の推定には、RANDOM ステートメントで出力される分散の期待値の構造より、1 次誤差成分と 2 次誤差成分を合成し、

$$\widehat{\text{Var}(R*A)} = (\widehat{\text{Var}}_{R*A} - \widehat{\text{Var}}_{\text{Error}}) / 4 = (0.7817 - 0.1150) / 4 = 0.1667$$

として計算することができる。A\*B の各水準の平均値の誤差 SE は、2 個のデータの平均なので

$$SE_{A*B} = \sqrt{(\widehat{\text{Var}}_{R*A} + \widehat{\text{Var}}_{\text{Error}}) / n_{A*B}} = \sqrt{(0.1150 + 0.1667) / 2} = 0.3753$$

となる。反復 R は、前節でのべた変量因子であるが、ここでは、"母数" とみなしたことになっている。

MIXED プロシジャーの表記法は、表示 7 のように、すべての因子を CLASS ステートメントにより指定し、MODEL ステートメントで母数因子を、RANDOM ステートメントで変量因子(Random Effects)を指定する。反復 R を RANDOM ステートメントではなく MODEL ステートメントに含めて母数因子としているが、これは従来の解析法に準じ

表示 6 分散の期待値の構造

Title2 ' <<< 反復がある 1 段分割実験 GLM プロシジャー >>> ' ; proc GLM data = DD.D2 ; /* REPB02A.TXT */ class R A B ; model Y = R A R*A B A*B / e1 ss1 ; random R*A ; test h = R A e = R*A ;																																																					
<b>分散分析表</b>																																																					
<table border="1"> <thead> <tr> <th>Source</th><th>DF</th><th>Type I SS</th><th>MS</th><th>F Value</th><th>Pr &gt; F</th></tr> </thead> <tbody> <tr> <td>R</td><td>1</td><td>5.0417</td><td>5.0417</td><td>43.84</td><td>0.0001</td></tr> <tr> <td>A</td><td>2</td><td>4.8533</td><td>2.4267</td><td>21.10</td><td>0.0004</td></tr> <tr> <td>R*A</td><td>2</td><td>1.5633</td><td>0.7817</td><td>6.80</td><td>0.0159 *</td></tr> <tr> <td>B</td><td>3</td><td>1.8683</td><td>0.6228</td><td>5.42</td><td>0.0210 *</td></tr> <tr> <td>A*B</td><td>6</td><td>6.8767</td><td>1.1461</td><td>9.97</td><td>0.0015 **</td></tr> <tr> <td>Error</td><td>9</td><td>1.0350</td><td>0.1150</td><td></td><td></td></tr> <tr> <td>Total</td><td>23</td><td>21.238</td><td></td><td></td><td></td></tr> </tbody> </table>						Source	DF	Type I SS	MS	F Value	Pr > F	R	1	5.0417	5.0417	43.84	0.0001	A	2	4.8533	2.4267	21.10	0.0004	R*A	2	1.5633	0.7817	6.80	0.0159 *	B	3	1.8683	0.6228	5.42	0.0210 *	A*B	6	6.8767	1.1461	9.97	0.0015 **	Error	9	1.0350	0.1150			Total	23	21.238			
Source	DF	Type I SS	MS	F Value	Pr > F																																																
R	1	5.0417	5.0417	43.84	0.0001																																																
A	2	4.8533	2.4267	21.10	0.0004																																																
R*A	2	1.5633	0.7817	6.80	0.0159 *																																																
B	3	1.8683	0.6228	5.42	0.0210 *																																																
A*B	6	6.8767	1.1461	9.97	0.0015 **																																																
Error	9	1.0350	0.1150																																																		
Total	23	21.238																																																			
<table border="1"> <thead> <tr> <th>Source</th><th>Type I Expected Mean Square</th></tr> </thead> <tbody> <tr> <td>R</td><td>Var(Error) + 4 Var(R*A) + Q(R)</td></tr> <tr> <td>A</td><td>Var(Error) + 4 Var(R*A) + Q(A, A*B)</td></tr> <tr> <td>R*A</td><td>Var(Error) + 4 Var(R*A)</td></tr> <tr> <td>B</td><td>Var(Error) + Q(B, A*B)</td></tr> <tr> <td>A*B</td><td>Var(Error) + Q(A*B)</td></tr> </tbody> </table>						Source	Type I Expected Mean Square	R	Var(Error) + 4 Var(R*A) + Q(R)	A	Var(Error) + 4 Var(R*A) + Q(A, A*B)	R*A	Var(Error) + 4 Var(R*A)	B	Var(Error) + Q(B, A*B)	A*B	Var(Error) + Q(A*B)																																				
Source	Type I Expected Mean Square																																																				
R	Var(Error) + 4 Var(R*A) + Q(R)																																																				
A	Var(Error) + 4 Var(R*A) + Q(A, A*B)																																																				
R*A	Var(Error) + 4 Var(R*A)																																																				
B	Var(Error) + Q(B, A*B)																																																				
A*B	Var(Error) + Q(A*B)																																																				
Tests of Hypotheses using the Type I MS for R*A as an error term																																																					
<table border="1"> <thead> <tr> <th>Source</th><th>DF</th><th>Type I SS</th><th>MS</th><th>F Value</th><th>Pr &gt; F</th></tr> </thead> <tbody> <tr> <td>R</td><td>1</td><td>5.0417</td><td>5.0417</td><td>6.45</td><td>0.1263</td></tr> <tr> <td>A</td><td>2</td><td>4.8533</td><td>2.4267</td><td>3.10</td><td>0.2436</td></tr> </tbody> </table>						Source	DF	Type I SS	MS	F Value	Pr > F	R	1	5.0417	5.0417	6.45	0.1263	A	2	4.8533	2.4267	3.10	0.2436																														
Source	DF	Type I SS	MS	F Value	Pr > F																																																
R	1	5.0417	5.0417	6.45	0.1263																																																
A	2	4.8533	2.4267	3.10	0.2436																																																

ているためで、本来は、変量因子とするべきだろう(変量とした場合は表示 14c)。1 次誤差 R\*A は 0.1667, 2 次誤差 Residual は 0.1150 と出力され手計算の結果と一致している。固定効果の F 検定も分散の期待値の構造に対応して行われている。

交互作用 A\*B が有意なので LSMEANS ステートメントにより、水準平均と誤差を求める。交互作用 A\*B の誤差は  $SE_{A*B} = 0.375$  と、適切に合成され、その 95% 信頼幅もオプションの指定により出力される。ただし、信頼幅を求めるための t 分布表の自由度はデフォルトでは、高次の誤差、この場合では、2 次誤差の自由度 9 を使ってしまい、信頼幅が常に小さめになり問題である。複数の誤差分散を合成した誤差の自由度の調整には、サタスウェイトの方法が知られているが、現在の SAS のリリース 6.07 では対応していない。そのために、実用的でないことは承知の上で、手計算で求めた調整した自由度 3.98 をオプションで指定して 95% 信頼幅を求めている。

ESTIMATE セットメントは、水準間の差、その誤差などを任意の線形式に対して求めることができる。因子 A 内での B の水準間の差、因子 B 内での A の水準間の差に興味がある。一例として  $A_1$  内の  $B_1$  と  $B_2$  水準の差、 $B_1$  内の  $A_1$  と  $A_2$  水準の差(この場合は自由度の調整が必要) および t 検定の結果を示す。

表示 7 MIXED プロシジャーによる出力例

```
Title2 '<<< 反復ある一段分割実験 R を母数 >>>';
proc MIXED data = DD.D2; /* REPB02E.TXT */
class R A B;
model Y = R A B A*B;
random R*A;
estimate 'A1:(B1 - B2)' B 1 -1 0 0
          A*B 1 -1 0 0 0 0 0 0 0 0 0 0;
estimate 'B1:(A1 - A2)' A 1 -1 0
          A*B 1 0 0 0 -1 0 0 0 0 0 0 0 / df=3.98;
lsmeans A*B / df=3.98 cl;
```

The MIXED Procedure  
Covariance Parameter Estimates (REML)

Cov Parm	Ratio	Estimate	Std Error	Z	Pr >  Z
R*A	1.4493	0.1667	0.1959	0.85	0.3949
Residual	1.0000	0.1150	0.0542	2.12	0.0339

Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F
R	1	2	6.45	0.1263
A	2	2	3.10	0.2436
B	3	9	5.42	0.0210 *
A*B	6	9	9.97	0.0015 **

Least Squares Means

Level	LSMEAN	Std Error	DDF	Lower	Upper
A*B A1 B1	12.0000	0.3753	3.98	10.9560	13.0440
A*B A1 B2	13.3000	0.3753	3.98	12.2560	14.3440
⋮					
A*B A3 B4	12.8000	0.3753	3.98	11.7560	13.8440

ESTIMATE Statement Results

Parameter	Estimate	Std Error	DDF	T	Pr >  T
A1:(B1 - B2)	-1.3000	0.3391	9	-3.83	0.0040 **
B1:(A1 - A2)	-2.7000	0.5307	3.98	-5.09	0.0147 *

#### 4. 整理された実験データの問題点

表示 5 のように実験のランダム化の過程が詳細に記述されていれば、適切な解析が可能であるが、表示 8 に示すように単に結果が整理（常に起きる問題である）されているとしよう。表を見ただけでは、繰り返しありの 2 元配置の実験が行なわれたかのように思われるかも知れない。しかしながら、ランダム化の手順により様々な分散分析の型があり、解析上は細心の注意を必要とする。

この実験には、3 水準の因子 A、4 水準の因子 B の 12 通りの組み合わせに対し、それぞれ 2 個、全部で 24 個のデータがある。さて、この表を見て全てのデータが、2 元配置分散分析の前提としている完全にランダム化された実験とみなして解析してよいのだろうか。GLM プロシジャーを使用したからといって結果が正しいという保証はまったく

ない。実際には差がないのに統計的に差があると言ったり、実際には差があるのに統計的に差がないと言ったり、誤用の結果はまちまちであり一定の傾向ではない。もちろん MIXED プロシジャーにも誤用の危険性が潜んでいることは、GLM プロシジャー以上であることを注意したい。そこで、それぞれの型に対応した分散分析表を示すと共に、それに対応する MIXED プロシジャーの使い方と結果の見方を次節以後に示す。

表示 8 整理された実験データ

	B1	B2	B3	B4
A1	12.3	13.9	15.0	15.0
A2	15.5	14.8	15.2	14.4
A3	13.9	13.3	14.3	13.6
	14.0	13.2	13.5	12.8

## 5. 完全なランダム化実験とみなした場合

繰り返しありの 2 元配置分散分析というのは、24 個のデータが、完全にランダム化されて得られたことを前提としているにもかかわらず、この前提が満たされるような実験は、現実的にはほとんど行われていないと考えたほうがよい。それでは、完全なランダム化実験とは、どのような実験なのだろうか。これは、表示 9 a に示すように、実際に実験データを得たランダム化の過程として考えるとよい。

表示 9 a 完全ランダム化の実験順序

	B1	B2	B3	B4
A1	(6) 12.3 (11) 11.7	(7) 13.9 (16) 12.7	(18) 15.0 (24) 13.0	(5) 15.0 (17) 13.4
A2	(19) 15.5 (21) 13.9	(1) 14.8 (4) 13.3	(3) 15.2 (10) 14.3	(8) 14.4 (23) 13.6
A3	(2) 13.9 (9) 14.0	(15) 13.3 (22) 13.2	(14) 14.3 (20) 13.5	(12) 12.8 (13) 12.8

括弧( )の中の数字が、実際に実験を行なう順番の一例である。第 1 番目の実験は A2, B2 の組み合わせで、第 2 番目の実験は A3, B1 の組み合わせで、…、第 24 番目の実験は A1, B3 の組み合わせで行なうのである。このような手順でほんとうに実験をしたならば、表示 9 b に示すように教科書的な繰り返しありの 2 元配置の分散分析を用いて何ら問題はない。この分散分析の結果は、因子 A の p 値が 0.0523 で統計的に差が有りそうに見えるが、有意水準  $\alpha = 0.05$  のを使うことを前提にすれば、有意な要因が全くないことが示されている。

表示 9 b 完全ランダム化の場合の分散分析表

要因	df	S	V	F <sub>0</sub>	p
A	2	4.8533	2.4267	3.81	0.0523 +
B	3	1.8683	0.6228	0.98	0.4353
A*B	6	6.8767	1.1461	1.80	0.1818
誤差	12	7.6400	0.6367		
全体	23	21.2383			

## 6. 測定の繰り返し

しばしば発生するのは、表示 10 a に示すように、実験が 12 回のランダム化しか行なわれなかつた場合である。第 1 回目の実験は A2, B2 の組み合わせで、その結果得られた 2 個のサンプルを、その場で測定して 14.8 と 13.3 が得られ、第 2 回目の実験は、A1, B3 の組み合わせで 15.0 と 13.0 が得られ、順次第 12 回目まで実験が行なわれたものとする。

なお、実験の結果として液体のようなサンプルを分割できるようなものは、それを 2 等分して測定しても同じ実験モデルと考えられる。さらに、24 個のサンプルを自動分析装置などで一括して測定しても同じ実験モデルが適用される。

表示 10 a 測定の繰り返しがある実験データ

	B1	B2	B3	B4
A1	(10) 12.3 11.7	(9) 13.9 12.7	(2) 15.0 13.0	(8) 15.0 13.4
A2	(4) 15.5 13.9	(1) 14.8 13.3	(6) 15.2 14.3	(12) 14.4 13.6
A3	(5) 13.9 14.0	(11) 13.3 13.2	(3) 14.3 13.5	(7) 12.8 12.8

実験の順序が記載されていないならば、表示 10 a のデータは、表示 9 a とまったく同じデータである。違いは、目には見えない各データに付いている誤差の構造が異なるのである。完全ランダム実験の場合は、各データにただ 1 つの誤差しか付いていないのに対し、この実験には 2 種類の誤差、実験誤差  $\epsilon^{(1)}$  と測定誤差  $\epsilon^{(2)}$  が、各データに付いている。

$$y_{111} = \mu + \alpha_1 + \beta_1 + (\alpha \beta)_{11} + \epsilon^{(1)}_{11} + \epsilon^{(2)}_{111} = 12.3$$

$$y_{112} = \mu + \alpha_1 + \beta_1 + (\alpha \beta)_{11} + \epsilon^{(1)}_{11} + \epsilon^{(2)}_{112} = 11.7$$

⋮

$$y_{342} = \mu + \alpha_3 + \beta_4 + (\alpha \beta)_{34} + \epsilon^{(1)}_{34} + \epsilon^{(2)}_{342} = 12.8$$

実験誤差  $\epsilon^{(1)}$  は、因子 A と因子 B の水準の組み合わせを変えて実験する度に付く誤差

である。 $y_{111}$  と  $y_{112}$  は、同時に実験されたのであるから共通の実験誤差  $\epsilon^{(1)}_{ij}$  が付き、それぞれのデータに別々の測定誤差  $\epsilon^{(2)}_{111}$  と  $\epsilon^{(2)}_{112}$  が付いている。問題は、交互作用  $(\alpha \beta)$  と実験誤差  $\epsilon^{(1)}$  は、それぞれのデータの添字が完全一致、言い換えれば完全に交絡しているので分離できない。このために、因子A および因子B の検定は、交互作用  $(\alpha \beta)$  と実験誤差  $\epsilon^{(1)}$  を合わせたものを誤差とみなして検定することになる。

分散分析の結果は、表示 10 b に示すように、分散 V の欄に検定の相手を明示するための矢印を入れておくとよい。因子A および因子B は、実験誤差  $\epsilon^{(1)}$  で検定しているので F 値は、表示 9 b の分散分析表の F 値に比べ小さくなり、因子A の水準間の差を全く捕えられなくなっている。

表示 10 b 測定の繰り返しがある場合の分散分析表

要因	df	S	V	F <sub>0</sub>	p
A	2	4.8533	2.4267	2.12	0.2015
B	3	1.8683	0.6228	0.54	0.6703
実験誤差 $\epsilon^{(1)}$	6	6.8767	1.1461	1.80	0.1818
測定誤差 $\epsilon^{(2)}$	12	7.6400	0.6367		
全体	23	21.2383			

さて、この実験計画では、何らかの差を見つけようとする試みは失敗してしまったのである。さらに、実験誤差あるいは交互作用すらも有意となっていないのは、実験誤差に比べて測定誤差が比較的大きいためであり、さらに実験を続けるのならば、測定の精度を高める工夫が必要がある、との教訓ぐらいは得られたことになる。

MIXED プロシジャーでの解析事例を表示 10 c に示す。RANDOMステートメントで交互作用 A\*B を誤差とみなすために指定することが特徴である。その結果、A\*B は、誤

表示 10 c 測定の繰り返しがある場合の MIXED プロシジャー

```
Title2 '"><<< 測定の反復 >>>' ;
proc MIXED data = DD.D2 ;
  class R A B ;
  model Y = A B ;
  random A*B ;
```

#### Covariance Parameter Estimates (REML)

Cov Parm	Ratio	Estimate	Std Error	Z	Pr >  Z
A*B	0.4000	0.2547	0.3555	0.72	0.4736
Residual	1.0000	0.6367	0.2599	2.45	0.0143

#### Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F
A	2	6	2.12	0.2015
B	3	6	0.54	0.6703

差とみなされ、因子Aと因子BのF値は、それぞれ2.12, 0.54と表示11bの分散分析表のF値と一致している。

## 7. 1段分割実験

表示11aに示すように、因子Aのランダム化と因子Bについてのランダム化が、階層的に行なわれることもある。実験は、ランダム化の手順により、まず因子A2の条件に固定し、その中で因子Bの水準をランダムな順序で実験を4回行ない、それぞれ15.5, 14.8, 15.2, 14.4が得られ、次に、因子A1の条件に固定した上で同様にランダムに実験を4回行ない12.3, 13.9, 15.0, 15.0を得、順次、因子Aの水準を順次変更して行く。

表示11a 1段分割実験のデータ

	ブロック	B1	B2	B3	B4
A1	R1 (2)	12.3	13.9	15.0	15.0
	R2 (3)	11.7	12.7	13.0	13.4
A2	R1 (1)	15.5	14.8	15.2	14.4
	R2 (5)	13.9	13.3	14.3	13.6
A3	R1 (4)	13.9	13.3	14.3	12.8
	R2 (6)	14.0	13.2	13.5	12.8

この実験の誤差の構造は、

$$y_{ijk} = \mu + \alpha_i + \epsilon^{(1)}_{ik} + \beta_j + (\alpha\beta)_{ij} + \epsilon^{(2)}_{ijk}$$

ただし、 $i=1, 2, 3$ ； $j=1, 2, 3, 4$ ； $k=1, 2$

のように、1次誤差 $\epsilon^{(1)}$ が因子Aの直後に現われ、交互作用 $(\alpha\beta)$ が1次誤差 $\epsilon^{(1)}$ と交絡しなくなっていることが、前節の測定の繰り返しの場合と異なる。表示11bに分散分析表を示す。

表示11b 1段分割実験の場合の分散分析表

要因	df	S	V	F <sub>0</sub>	p
A 1次誤差 $\epsilon^{(1)}$	2	4.8533	2.4267	1.10	0.4377
	3	6.6050	2.2017	19.14	0.0003 ***
B	3	1.8683	0.6228	5.42	0.0210 *
A*B	6	6.8767	1.1461	9.97	0.0015 **
2次誤差 $\epsilon^{(2)}$	9	1.0350	0.1150		
全体	23	21.2383			

MIXED プロシジャーでの解析事例を表示11cに示す。表示11aのブロックRの水準は、単なる実験順序であり、対応があるわけではない。従って、RANDOMステートメントで1次誤差をR(A)と指定する。この表記法は、GLM プロシジャーでの考え方と同じで、

対応のない変量因子Rと交互作用 R\*Aを合わせたものをまとめて誤差とする記述方式であり、A(R)としても同じであるが、誤差であることを意識して R(A)としている。

表示 11c 1段分割実験に対する MIXED プロシージャ

Title2 '"><<< 一段分割 >>>' ; proc MIXED data = DD.D2 ; /* REPB02C.TXT */ class R A B ; model Y = A B A*B ; random R(A) ;																										
Covariance Parameter Estimates (REML)																										
<table border="1" style="width: 100%; border-collapse: collapse;"> <thead> <tr> <th>Cov Parm</th><th>Ratio</th><th>Estimate</th><th>Std Error</th><th>Z</th><th>Pr &gt;  Z </th></tr> </thead> <tbody> <tr> <td>R(A)</td><td>4.5362</td><td>0.52177</td><td>0.4496</td><td>1.16</td><td>0.2459</td></tr> <tr> <td>Residual</td><td>1.0000</td><td>0.11500</td><td>0.0542</td><td>2.12</td><td>0.0339</td></tr> </tbody> </table>							Cov Parm	Ratio	Estimate	Std Error	Z	Pr >  Z	R(A)	4.5362	0.52177	0.4496	1.16	0.2459	Residual	1.0000	0.11500	0.0542	2.12	0.0339		
Cov Parm	Ratio	Estimate	Std Error	Z	Pr >  Z																					
R(A)	4.5362	0.52177	0.4496	1.16	0.2459																					
Residual	1.0000	0.11500	0.0542	2.12	0.0339																					
Tests of Fixed Effects																										
<table border="1" style="width: 100%; border-collapse: collapse;"> <thead> <tr> <th>Source</th><th>NDF</th><th>DDF</th><th>Type III F</th><th>Pr &gt; F</th></tr> </thead> <tbody> <tr> <td>A</td><td>2</td><td>3</td><td>1.10</td><td>0.4377</td></tr> <tr> <td>B</td><td>3</td><td>9</td><td>5.42</td><td>0.0210 *</td></tr> <tr> <td>A*B</td><td>6</td><td>9</td><td>9.97</td><td>0.0015 **</td></tr> </tbody> </table>							Source	NDF	DDF	Type III F	Pr > F	A	2	3	1.10	0.4377	B	3	9	5.42	0.0210 *	A*B	6	9	9.97	0.0015 **
Source	NDF	DDF	Type III F	Pr > F																						
A	2	3	1.10	0.4377																						
B	3	9	5.42	0.0210 *																						
A*B	6	9	9.97	0.0015 **																						

この結果は、これまでの分散分析の結果と大きく違ひ、因子Bの主効果、交互作用 A\*Bともに統計的に有意な差となっている。どうしてなのであろうか。これは、因子Aの水準の変更に伴い、かなり大きな1次誤差  $\epsilon^{(1)}$  があるにもかかわらず、その中の因子Bの変更に伴って発生する2次誤差  $\epsilon^{(2)}$  が相対的に小さいからである。それは、表示 12 に示すように、A1 の R1 のデータと R2 のデータを比べてみると、R2 のデータの方が常に大きく、次に、それぞれの R1, R2 内のデータ 4 個の平均値からの偏差 d を計算する。この d の差が、y の差に比べて相対的に小さい（2次誤差  $\epsilon^{(2)}$  に相当する）ことがわかる。このことは、因子Aの水準の変更に伴う誤差が大きく、因子Aの水準内での因子Bの水準の変更に伴う誤差が小さいことを意味し、因子Bの主効果と交互作用 A\*B が統計的に有意な差となったと解釈される。

表示 12 1次誤差と2次誤差の構造

	A1 内 R1		A1 内 R2		A1 内の差 R1 - R2	
	$y_{1j1}$	偏差 $d_{1j1}$	$y_{1j2}$	偏差 $d_{1j2}$	y の差	d の差
B1	12.3	-1.75	11.7	-1.00	0.60	-0.75
B2	13.9	-0.15	12.7	0.00	1.20	-0.15
B3	15.0	+0.95	13.0	+0.30	2.00	0.65
B4	15.0	+0.95	13.4	+0.70	1.60	0.20
平均	14.05		12.70			

## 8. 実験の反復

繰り返しを含むような実験を企画する場合は、完全なランダム化よりも、繰り返しのない要因配置実験を反復する方が、いろいろな意味で実験計画上好ましい性質を持つ場合が多い。ある実験が、1週間で12回の実験しか物理的にできないような場合に、全体で24回の実験を2週間でランダム化するよりも、1週間単位で12回のランダム化を2回反復すれば、実験時期に起因するさまざまなもの、実験の場の変化を別枠で取り出すことが可能となり、実験の精度を高めることができる。

具体的には、表示13aに示すように第1反復をR1として、最初の12回の実験を因子Aと因子Bの全ての組み合わせに対しランダムに行なう。次に、第2反復をR2として、(13)から(24)までのランダム化を行なう。もしも、第3反復を行なうのであるならば、さらに(25)から(36)と実験を追加すればよい。

データの構造は、反復Rを $\gamma_k$ ,  $k=1, 2$ とし

$$y_{kij} = \mu + \gamma_k + \epsilon^{(0)}_k + \alpha_i + \beta_j + (\alpha\beta)_{ij} + \epsilon^{(1)}_{kij}$$

となる。因子Aおよび因子Bは、 $\epsilon^{(1)}$ で検定し、反復Rに伴う誤差 $\epsilon^{(0)}$ は、 $\gamma$ と交絡しているので、分離して取り出すことはできない。

表示13a 反復実験のデータ

反復		B1		B2		B3		B4	
		R1	A1	(10)	12.3	(9)	13.9	(2)	15.0
	A2	(4)	15.5	(1)	14.8	(6)	15.2	(12)	14.4
	A3	(5)	13.9	(11)	13.3	(3)	14.3	(7)	12.8
R2	A1	(15)	11.7	(21)	12.7	(13)	13.0	(18)	13.4
	A2	(20)	13.9	(16)	13.3	(22)	14.3	(17)	13.6
	A3	(23)	14.0	(14)	13.2	(19)	13.5	(24)	12.8

表示13aは、一見すると繰り返しのない3元配置分散分析の型かの様に見える。このままで単純に分散分析をすると、反復Rと他の因子との交互作用 $R*A$ ,  $R*B$ が発生し、3因子交互作用 $R*A*B$ を誤差項として検定することになる。これは完全な誤りで、交互作用 $R*A$ は、因子Aの反復R1の水準平均と反復R2の水準平均の差の食い違いの大きさを示すものであって、これも実験誤差と考えるべきものである。同様に、 $R*B$ も誤差であり、 $R*A*B$ に合わせて実験誤差 $\epsilon^{(1)}$ とみなしそれを表示13bに示すような分散分析表を作成する。

この結果から、反復Rの分散が、5.0417と誤差分散0.2362に比べ極めて大きいことから実験の場の予期せぬ変化があったことになり、それにもかわらず、その中では精度の高い実験が行なわれていたことが、因子Aおよび交互作用 $A*B$ が高度に有意な差と

表示 13b. 反復実験の場合の分散分析表

要因	df	S	V	F <sub>0</sub>	p
R	1	5.0417	5.0417	21.34	0.0007 ***
A	2	4.8533	2.4267	10.27	0.0030 **
B	3	1.8683	0.6228	2.67	0.1019
A*B	6	6.8767	1.1461	4.85	0.0117 *
誤差 e <sup>(1)</sup>	11	2.5983	0.2362		
全体	23	21.2383			

なっているのである。

MIXED プロシジャーでの解析事例を表示 13c に示す。反復 R のみを、RANDOMステートメントで指定しているが、表示 13b の分散分析表は、反復 R を入れ "母数" とみなした場合となっている。分散分析表までの段階では、"変量" とみなした場合どちらでも母数因子についての検定結果は同じである。違いは、第 2 節の変量モデルで示したように、LSMEANS ステートメントなどで標準誤差を推定するときに異なる。

表示 13c 反復実験に対する MIXED プロシジャー

```
Title2 '"><<< 反復実験 >>>';  
proc MIXED data = DD.D1; /* REPB02D.TXT */  
  class R A B;  
  model Y = A B A*B;  
  random R;
```

#### Covariance Parameter Estimates (REML)

Cov Parm	Ratio	Estimate	Std Error	Z	Pr >  Z
R	1.6953	0.4004	0.5942	0.67	0.5004
Residual	1.0000	0.2362	0.1007	2.35	0.0190

#### Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F
A	2	11	10.27	0.0030 **
B	3	11	2.64	0.1019
A*B	6	11	4.85	0.0117 *

## 9. 反復ある1段分割実験データ

前節で述べた反復実験は、優れた実験計画はであるが、ランダム化の範囲がまだ大きい。これに第 7 節で述べた分割実験と組み合わせてみよう。表示 14a に実験順序の例を示す。ランダム化は、反復 R1 の中で因子 A の水準変更を行ない、その中で、さらに因子 B の 4 水準の中でランダム化を行なっている。反復 R2 でも同様なランダム化を行なう。なお、この実験計画は、表示 5 で示した分割実験のデータと全く同じであり、ランダム化の順序を加えたものである。

表示 14a 反復ある 1 段分割実験データ

反復		B1	B2	B3	B4	平均
R1	A1 (2)	12.3	13.9	15.0	15.0	14.05
	A2 (1)	15.5	14.8	15.2	14.4	14.98
	A3 (3)	13.9	13.3	14.3	12.8	13.58
R2	A1 (5)	11.7	12.7	13.0	13.4	12.70
	A2 (6)	13.9	13.3	14.3	13.6	13.78
	A3 (4)	14.0	13.2	13.5	12.8	13.38

データの構造は、

$$y_{kij} = \mu + \gamma_k + \epsilon^{(0)}_k + \alpha_i + \epsilon^{(1)}_{ki} + \beta_j + (\alpha \beta)_{ij} + \epsilon^{(2)}_{kij}$$

となり、因子Aのあとに2次誤差  $\epsilon^{(1)}$  が付くことが、前節の反復実験と異なる。この誤差は、3配置分散分析とみなした場合の交互作用 R\*A に相当する。表示 6 の GLM プロシジャーの出力を整理すると表示 14 b の分散分析表となる。

表示 14 b 反復のある 1 段分割実験の分散分析表

要因	df	S	V	F <sub>0</sub>	p
R	1	5.0417 ↘	5.0417	6.45	0.1263
A	2	4.8533 ↘	2.4267	3.10	0.2436
誤差 $\epsilon^{(1)}$	2	1.5633 ↘	0.7817	6.80	0.0159 *
B	3	1.8683 ↘	0.6228	5.42	0.0210 *
A*B	6	6.8767 ↘	1.1461	9.97	0.0015 **
誤差 $\epsilon^{(2)}$	9	1.0350 ↘	0.1150		
全体	23	21.2383			

因子Aは、1次誤差  $\epsilon^{(1)}$  で検定し、因子Bと交互作用 A\*B は、2次誤差  $\epsilon^{(2)}$  で検定する。1次誤差  $\epsilon^{(1)}$  の分散 0.7817 は、2次誤差  $\epsilon^{(2)}$  の分散 0.1150 と比較し有意に大きい。これは、因子Aの各水準の反復間の平均値の差、

$$\begin{aligned} A1 &: 14.05 - 12.70 = 1.35 \\ A2 &: 14.98 - 13.78 = 1.20 \\ A3 &: 13.58 - 13.38 = 0.20 \end{aligned}$$

特に A3 が他の水準に比べかなり小さく、因子Bの水準変更に伴う 2次誤差  $\epsilon^{(2)}$  に対しに比べて 1次誤差  $\epsilon^{(1)}$  大きいことの原因となっている。

反復 R を変量とみなした場合の MIXED プロシジャーによる解析事例を表示 14 c に示す。反復 R と 2次誤差 R\*A を、RANDOMステートメントで指定する。LSMEANS ステートメントで推定される交互作用 A\*B の標準誤差は、0.5642 となっている。これは、誤差の分散成分の推定値

Var(R)	0.3550
Var(R*A)	0.1667
Var(Error)	0.1150
計	0.6367

を加え合わせ、

$$SE_{A*B} = \sqrt{(\text{誤差} / n_{A*B})} = \sqrt{(0.6367 / 2)} = 0.5642$$

として計算したものと一致している。反復 R を"母数"とみなした表示 7 の  $SE_{A*B}=0.3753$  に比べ Var(R) の分だけ大きくなっていることがわかる。

表示 14 c 反復を変量とした 1 段分割実験に対する MIXED プロシジャー

```
Title2 '"><<< 反復ある一段分割実験 R を変量 >>>';  
proc MIXED data = DD.D2 ; /* REPB02E.TXT2 */  
  class R A B ;  
  model Y = A B A*B ;  
  random R R*A ;  
  lsmeans A*B ;
```

#### Covariance Parameter Estimates (REML)

Cov Parm	Ratio	Estimate	Std Error	Z	Pr >  Z
R	3.0870	0.3550	0.5977	0.59	0.5526
R*A	1.4493	0.1667	0.1959	0.85	0.3949
Residual	1.0000	0.1150	0.0542	2.12	0.0339

#### Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F
A	2	2	3.10	0.2436
B	3	9	5.42	0.0210 *
A*B	6	9	9.97	0.0015 **

#### Least Squares Means

Level	LSMEAN	Std Error	DDF	T	Pr >  T
A*B A1 B1	12.0000	0.5642	9	21.27	0.0000
A*B A1 B2	13.3000	0.5642	9	23.57	0.0000
A*B A3 B4	12.8000	0.5642	9	22.69	0.0000

## 10. 反復のある 2 方分割実験

これまで、因子 A を第 1 分割とした例を示してきたのであるが、因子 B についても第 1 の分割の対象ととしてもよい。どちらかは、実際の実験のやりやすさに依存する。

表示 15 a に示すように因子 A と因子 B を同時に分割の対象とすることができる。これは、2 方分割と呼ばれているもので、反復 R1 の中で因子 A の水準変更を行い、第 1 回目は A2 水準であるが、この時にまとめて 4 個の実験サンプルを作つておく。次に A1 水準でまとめて 4 個の実験サンプルを、さらに A3 水準で 4 個の実験サンプルを作り、合計 12 個の実験サンプルを作る（4 個の実験サンプルでなく 1 個の実験サンプルを作り、

後で4等分してもよい). 因子Bでは、因子Aのそれぞれの水準で作成された4個の実験サンプルより1個づつ抜き出し、合計3個のサンプルに対し、まとめて実験して結果を得るデータの構造は、

$$y_{kij} = \mu + \gamma_k + \epsilon^{(0)}_k + \alpha_i + \epsilon^{(1a)}_{kij} + \beta_j + \epsilon^{(1b)}_{kij} + (\alpha\beta)_{ij} + \epsilon^{(2)}_{kij}$$

となり、因子Bのあとに誤差 $\epsilon^{(1b)}$ が付くことが、前節の反復実験と異なる。この誤差は、3元配置分散分析とみなした場合の交互作用 $R*B$ に相当する。

表示 15 a 反復のある2方分割実験データ

		B1	B2	B3	B4
R1	A1 (2)	[3]	[2]	[4]	[1]
		12.3	13.9	15.0	15.0
		A2 (1)	15.5	14.8	15.2
R2	A3 (3)	13.9	13.3	14.3	12.8
		[7]	[5]	[8]	[6]
		A1 (6)	11.7	12.7	13.0
	A2 (5)	13.9	13.3	14.3	13.6
		A3 (4)	14.0	13.2	13.5
					12.8

分散分析表を、表示 15 b に示す。因子Aはa方向の1次誤差 $\epsilon^{(1a)}$ で、因子Bはb方向の1次誤差 $\epsilon^{(1b)}$ で、交互作用 $A*B$ は2次誤差 $\epsilon^{(2)}$ でそれぞれ検定する。反復Rは誤差 $\epsilon^{(1a)}$ と誤差 $\epsilon^{(1b)}$ を合成したもので検定する必要がある。検定結果から、1次誤差 $\epsilon^{(1b)}$ のF値は、1.00以下であり、一般的には2次誤差にプールして分散分析をやり直す。その結果は、表示 14 b に示した反復のある1段分割実験の分散分析表に帰着する。

表示 15 b 反復のある2方分割の分散分析表

要因	df	S	V	F <sub>0</sub>	p
R	1	5.0417	5.0417	-	-
A 1次誤差 $\epsilon^{(1a)}$	2	4.8533	2.4267	3.10	0.2436
	2	1.5633	0.7817	5.91	0.0510+
B 1次誤差 $\epsilon^{(1b)}$	3	1.8683	0.6228	7.73	0.0635+
	3	0.2417	0.0806	0.61	0.6332
A*B 2次誤差 $\epsilon^{(2)}$	6	6.8767	1.1461	8.67	0.0094 **
	6	0.7933	0.1322		
全体	23	21.2383			

表示 15 b の分散分析表を適切に構成するためには、GLM プロシジャーの RANDOM ステートメント、または、VARCOMP プロシジャーにより、事前に分散の期待値の構造を確認しておくことが望ましい。MIXED プロシジャーとの関連から言えば、表示 15c に示すように VARCOMP プロシジャーの使用を薦める。この結果で注意してもらいたいのは、

$b$  方向の 1 次誤差  $\epsilon^{(1b)}$ ,  $\text{Var}(R(B))$  の推定量が, -0.0172 とマイナスになっていることがある。これは、形式的に,

$$\text{Var}(R(A)) = (0.0806 - 0.1323) / 3 = -0.0172$$

と計算している結果であり、1 次誤差  $\epsilon^{(1b)}$  の F 値は、1.00 以下であることに対応している。

表示 15 c 反復のある 2 方分割の分散の期待値の構造

```
Title2 '"><<< 2 方分割実験 >>>' ;
proc VARCOMP data = DD.D2 method=TYPE1 ; /* REPB02F.TXT */
  class R A B ;
  model Y = A B A*B R R*A R*B / fixed=3 ;
```

Variance Components Estimation Procedure

Source	DF	Type I SS	Type I MS	Expected Mean Square
A	2	4.8533	2.4267	$\text{Var}(\text{Error}) + 4 \text{Var}(R^*A) + Q(A, A^*B)$
B	3	1.8683	0.6228	$\text{Var}(\text{Error}) + 3 \text{Var}(R^*B) + Q(B, A^*B)$
A*B	6	6.8767	1.1461	$\text{Var}(\text{Error}) + Q(A^*B)$
R	1	5.0417	5.0417	$\text{Var}(\text{Error}) + 3 \text{Var}(R^*B) + 4 \text{Var}(R^*A) + 12 \text{Var}(\text{Error})$
R*A	2	1.5633	0.7817	$\text{Var}(\text{Error}) + 4 \text{Var}(R^*A)$
R*B	3	0.2417	0.0806	$\text{Var}(\text{Error}) + 3 \text{Var}(R^*B)$
Error	6	0.7933	0.1322	$\text{Var}(\text{Error})$
Total	23	21.2383		

  

Variance Component	Estimate
$\text{Var}(R)$	0.3593
$\text{Var}(R^*A)$	0.1624
$\text{Var}(R^*B)$	-0.0172
$\text{Var}(\text{Error})$	0.1322

さて、MIXED プロシージャは、2 方分割の問題に適切に対処できるか、が今回の検討のメインテーマであった。これまでの結果から、このデータでは、 $b$  方向の 1 次誤差  $\epsilon^{(1b)}$  の存在は否定されているのであるが、あえて、表示 15 d に示すように、2 方分割の問題のままで解析した。まず、分散成分であるが、 $\widehat{\text{Var}}(R^*B) = 0.00$  とし、論理的にありえないマイナスになることを避けている。次に、 $\widehat{\text{Var}}(\text{Error}) = 0.1150$  は、表示 15 c の 0.1322 と異なる。これは、 $\widehat{\text{Var}}(R^*B) = 0.00$  としたことに伴い、 $R^*B$  の平方和を誤差の平方和に自動的にプールして分散を計算し直していることによる。

$$V_e = (S_{R^*B} + S_e) / df_{R^*B+e} = (0.2417 + 0.7933) / (3 + 6) = 0.1150$$

これは、表示 14 c の反復のある 1 段分割実験の分散分析の結果となっている。母数因子 Fixed Effects の因子 A の p 値は、0.2436 で一致しているが、因子 B と交互作用 A\*B の F 値 5.42, 9.97 は、表示 14 b の結果と一致しているものの、p 値は、0.0994 と 0.0066 と一致しない。これは、DDF に出力されている F 分布表の分母の自由度が異なるからである。これは、分散は合成したものを使用しているにもかかわらず、F 分布表の分母の

自由度は、元のモデルのままの DDF = 6 を使用していることに原因がある。いずれにしても、あるレベルの誤差分散が、統計的に無いと判断された場合には、解析モデルを変更することが望ましい。ただし、有意でない母数因子を誤差にプーリングする場合には、言うまでもないが、あらかじめその方式を決めておくべきであり、結果を見てから色々と試行錯誤するべきではない。

表示 15 d 反復のある 2 方分割に対する MEXED プロシジャー

```
Title2 '<<< 2 方分割実験 >>>';
proc MIXED data = DD.D2 ;
  class R A B ;
  model Y = A B A*B ;
  random R R*A R*B ;
```

#### Covariance Parameter Estimates (REML)

Cov Parm	Ratio	Estimate	Std Error	Z	Pr >  Z
R	3.0870	0.3550	0.5977	0.59	0.5526
R*A	1.4493	0.1667	0.1959	0.85	0.3949
R*B	0.0000	0.0000			
Residual	1.0000	0.1150	0.0542	2.12	0.0339

#### Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F
A	2	2	3.10	0.2436
B	3	3	5.42	0.0994 +
A*B	6	6	9.97	0.0066 **

## 11.まとめ

分散分析は、実験計画法に従った実験データの解析の基本であり、多くの統計の教科書で取り上げられ、大小様々な統計パッケージにも必ず含まれている。しかしながら、その多くは完全にランダム化された理想の実験データの解析を主体にしており、現実の実験の場でしばしば無意識の内に行われているランダム化が限定された分割実験については、ほとんど無視されている。人および動物に対する実験の場では、必然的に起きていることの多い変量因子の取り扱いについても無視されがちである。

その結果として、分散分析の誤用がしばしば発生するのである。交互作用を無視して主効果のみを云々したり、変量因子の認識が全く欠如していたり、明らかな分割実験をしているにも関わらず完全にランダム化したとみなしたり、実に様々である。SAS の GLM プロシジャーは、多様な実験計画モデルにたいし適用できる優れた統計パッケージではあるが、変量因子を含んだり、あるいは分割実験のように必然的に複数の誤差をもつ様な計画に対しては、本質的な解決がされている訳ではない。

新たにリリースされた MIXED プロシジャーは、まさに、これらの問題を本格的に取り扱えるプロシジャーである。これまで、計算が煩雑であるがゆえに敬遠されがちな問題を

ランダム化の方法という観点から整理してみた。分割実験には様々な型があり、同じデータであっても些細なと思われるような取扱いの違いが、全く異なる事例を、MIXED プロシージャの基本的な使い方、結果の解釈のしかたと共に示してきた。

MIXED プロシージャの出力には、従来の伝統的な分散分析表の F 値のみが出力され、平方和および分散は出力されない。このために、その F 値が妥当なのか、結果を見ただけでは判断しがたい。微妙な使い方の違いが、大きな結果の違いに結び付くだけに問題である。そのために、分散の期待値の構造を把握することが肝要であり、現在は VARCOMP プロシージャあるいは GLM プロシージャを併用しながら慎重に対処する必要がある。

MIXED プロシージャは固定効果についてセル平均モデル Type III SS をに基づく Type III の F 値の出力がなされているが、VARCOMP プロシージャでの分散の期待値の構造の出力は、Type I の平方和に基づくもののみであり、GLM プロシージャでは、Type I, Type II, Type III の分散の期待値の構造の出力がされ、MIXED プロシージャとの整合性の面では、Type III が望ましい。ただし、データがバランスが取れていれば、どちらでも同じなのでここでは Type I を示した。違いは、データがアンバランスな場合に起き、この場合にはもはや GLM プロシージャで出力される分散の期待値の構造が複雑になり正確な検定を手作業で行うのはかなり面倒であり、VARCOMP プロシージャの Type I の場合は、求められた分散成分の推定値をもとにして、面倒ではあるが手計算可能ではあるが実用的とは言えない。

なお、今回の報告で含めなかった直交表を用いた分割実験の場合は、GLM プロシージャでは、Type I のでしか解析できないが、MIXED プロシージャでは、Type III で対応でき、アンバランスな場合に適切な検定が可能である。

さらに、MIXED プロシージャは、経時測定データへの応用も大いに期待できるのであるが、別の機会に、様々な活用方法を今回と同様に GLM プロシージャと対比しながら報告したい。

## 文献

- 1) SAS Institute Inc. (1992) : Technical Report P229, SAS / STAT Software Change and Enhancement, Cary NC USA: SAS Institute.
- 2) 高橋行雄, 大橋靖雄, 芳賀敏郎 (1990) : SAS による実験データの解析, 東大出版会.
- 3) 山内二郎ほか編 (1977) : 品質管理便覧 7.3.1 実験計画モデルの問題, 日本規格協会.
- 4) 中里博明 (1970) : 分割実験法入門, 日科技連出版社.
- 5) 朝尾正, 安藤貞一, 楠木正, 中村恒夫 (1973) : 最新実験計画法, 日科技連出版社.
- 6) 田中豊 (1985) : パソコン実験計画法, 現代数学社.
- 7) Box, G.P.E., Hunter, W.G. and Hunter, J.S. (1978) : Statistics for Experimental, John Wiley & Sons.
- 8) 豊田秀樹 (1994) : 違いを見ぬく統計学 - 実験計画と分散分析 -, 講談社.